
1 Circuit Simulation with SPICE 
Circuit simulation is one of the fundamental steps in the design methodology of VLSI circuits.  The 
operation characteristics of the designed product are predicted in a cost and time efficient manner 
using simulation.  In the current assignment, circuit simulation is used to illustrate basic concepts of 
digital circuits and to learn how to read and write SPICE files. SPICE is the most commonly used 
analog circuit simulator today and is enormously important for the electronics industry. SPICE is a 
general purpose analog simulator which contains models for most circuit elements and can handle 
complex nonlinear circuits. 

1.1 Circuit elements 
In SPICE syntax, all elements within a circuit are labeled as components.  The wires, for instance, 
are labeled as nets, and the connectivity of the circuit is defined via these nets.  The standard 
SPICE syntax requires the definition of an element (labeled by a letter such as R for a resistor, C for 
a capacitor etc.), the nets its terminals are connected to, and the value for the element.  The SPICE 
syntax for component definitions looks as follows: 
 
C0 gnd net1 10f            
R0 net2 net3 1k            
V3 net4 net5 DC 3.3V 
V0 net6 net7 pulse 0 vddp 0n 200p 200p 3n 6n  
 
 The first line is a capacitor, named C0, between gnd and net1.  Its value is specified to be 
10fF.  The second line is a 1K resistor, called R0, connected between net2 and net3.  The third line is 
a 3.3V DC voltage source named V3, between net4 and net5.  The fourth line is another voltage 
source, connected between net6 and net7.  The voltage source produces pulses of 0 V and vddp (e.g. 
5V).  The time values listed are respectively delay, rise_time, fall_time, pulse_width 
and period. 
 
The format for a MOSFET is: 
 

1. Name (must begin with M) 

2. Node numbers in the following order : Drain Gate Source Base 

3. MOSFET model name 

4. W and L are the width and length of the gate (in meters) [μ is micron] 

 
For example,  
 
M_0 output input vdd vdd p W=2u L=1u 



M_1 output input gnd gnd n W=1u L=1u 
 
Note that the first letter of each line in the above examples specifies the element type.  Most popular 
elements, used in this course are: 
 
V Voltage 
C Capacitance 
R resistance 
M MOSFET 
Q BJT 

1.2 Voltage input waveforms 
There are some commands to insert input waveforms in the circuit, as well.  ‘PWL’ and ‘PULSE’ are 
two input waveforms frequently used in transient analysis.  The format for a piece-wise linear (PWL) 
waveform is: 
 
PWL time0 value0 time1 value1 ... 
 
where value0 is the value at time0, value1 is the value at time1 and so on. 
 An example could be: 
 
Vin 3 0 PWL 0n 0V 0.4n 5V 14.6n 5V 15n 0V 
 
 The format for the PULSE command is: 
 
PULSE (initial_value pulse_value delay rise_time fall_time 
pulse_width period) 
 
 
An example for this input waveform command is: 
 
Vin 3 0 PULSE (0 5 4.9n 0.2n 0.2n 4.8n 10n) 

1.3 Common commands 
The syntax to define circuit components are listed in the previous sections.  Here are some basic 
HSPICE commands: 

• .INCLUDE is used to include text from one file at run time. 
• .PARAM is used to define a parameter that is used as a variable in other statements in the 

netlist file. 
• .DC performs DC analysis.  For example .DC Vin 0 5 0.1 performs a DC analysis 

using a sweep of Vin from 0V to 5V in 0.1V increments. 
• .TRAN performs a transient analysis. For example, .TRAN 0.1n 50n. 
• .IC is useful for setting initial conditions for transient solutions.  This is essential 

whenever the circuit stores information, such as latches, flip-flops or on dynamic storage 
nodes (capacitances etc.).  For example, .IC V(1)=5 initializes node 1 to 5V 

• .END is required at the end of an HSPICE netlist file. 



1.4 Subcircuits 
A subcircuit allows you to define a collection of elements as a subcircuit (e.g. an RC circuit) and to 
insert this description into the overall circuit. A subcircuit definition is begun by a .SUBCKT, 
followed by the circuit description as follows: 
 
.SUBCKT SUBNAME N1 N2 N3 …… 
Element statements 
. 
. 
. 
.ENDS SUBNAME 
 
where SUBNAME is the subcircuit name and N1, N2, N3 are the external nodes of the subcircuit. 
The external nodes cannot be 0. The node numbers used inside the subcircuit are strictly local, with 
the exception of 0 (ground) which is always global. .ENDS must be used at the end of the subcircuit 
definition. The SUBNAME (subcircuit name) used with the .ENDS statement indicates which 
subcircuit definition is being terminated. If SUBNAME is omitted, all subcircuits being defined are 
terminated. The name is needed only when nested subcircuit definitions are being made. 
 
The following example shows the use of a subcircuit: 
 
.SUBCKT RC_Circuit net1 net2 
R1 net1 net2 1k 
C1 net2 gnd 10f 
.ENDS RC_Circuit 
 
 
X1 vdd netx RC_Circuit 
 
When using a subcircuit, its name must start with ‘X’. In the above example, the subcircuit is called 
X1. 
 

1.5 Simulation 
In order to run simulations on the netlist file, specify the durations of the simulation and/or the initial 
and incrementing values for the input voltages.  Basically two types of simulations will be used: 
 

1. DC Sweep:  The .DC statement allows you to increment (sweep) an independent source 
over a certain range with a specified step value. The format of using this statement is as 
follows: 

 
.DC source_name Vstart Vstop Vincrement 
 
The DC sweep command can be nested and is often used to plot transistor characteristics. 
The format for the nested DC Sweep is: 
 
.DC source_name1 Vstart1 Vstop1 Vincrement1 source_name2 
Vstart2 Vstop2 Vincrement2 
 
 



2. Transient Analysis: The .TRAN statement specifies the time interval over which the 
transient analysis takes place and the time increment. The format of this command is: 

 
.TRAN tstep tstop <tstart <tmax>> 
 
where tstep is the time increment, tstop is the final time, tstart is the starting time (if 
omitted, tstart is assumed to be zero), tmax is the maximum step size. tstart and tmax are 
optional.  
For example :  
  
.TRAN 0.01N 20N 

1.6 SPICE code 
The SPICE file should contain the technology parameters, in order to produce a good prototype of 
the circuit and produce realistic simulation results.  This is done by including a technology file stored 
in a path, or by directly trying to write the parameters.  Most of the manufacturers release their 
process technology parameters for use in simulation and circuit realization. 
 
 Now look at the following source file: 
 
* Steve test 
simulator lang=spectre 
global 0 
include "$CDK_DIR/models/hspice/public/tsmc18dN.m" 
include "$CDK_DIR/models/hspice/public/tsmc18dP.m" 
 
// sub circuit for inverter 
subckt INV1 A gnd out vdd 
    M1 (out A gnd gnd) tsmc18dN w=540.0n l=180.0n as=2.43e-13 \ 
        ad=2.43e-13 ps=1.98u pd=1.98u m=1 region=sat 
    M2 (out A vdd vdd) tsmc18dP w=1.08u l=180.0n as=4.86e-13 \ 
        ad=4.86e ps=3.06u pd=3.06u m=1 region=sat 
ends INV1 
 
// the main circuit 
I0 (A 0 INV net9) INV1 
V0 (net9 0) vsource type=dc dc=1.8 
V1 (A 0) vsource type=pulse val0=0 val1=1.8 period=1n \  
           delay=0 rise=10p fall=10p width=490p 
 
//Instructions to Spectre (some unique to spectre vs spice) 
simulatorOptions options reltol=1e-3 vabstol=1e-6 \ 
iabstol=1e-12 temp=27 tnom=27 scalem=1.0 scale=1.0 gmin=1e-12 \ 
rforce=1 maxnotes=5 maxwarns=5 digits=5 cols=80 pivrel=1e-3 \ 
sensfile="../psf/sens.output" checklimitdest=psf 
 
//The request for a transient analysis 
tran tran stop=20n write="spectre.ic" writefinal="spectre.fc" \ 
    annotate=status maxiters=5 
 
// The commands to put the results in the “rawfile” directory 



finalTimeOP info what=oppoint where=rawfile 
modelParameter info what=models where=rawfile 
element info what=inst where=rawfile 
outputParameter info what=output where=rawfile 
designParamVals info what=parameters where=rawfile 
primitives info what=primitives where=rawfile 
subckts info what=subckts  where=rawfile 
saveOptions options save=allpub 
 

• This file is very similar to the file that Virtuoso should generate for your inverters.  
• The lines starting with ‘*’ are just comments and not a part of the code. In Spectre ‘//’ starts 

a comment as well.  
• The next line tells Spectre to use “Spectre style” commands. The most obvious difference is 

that Spectre does not start is commands with a “.” You can ask Spectre to use SPICE style 
commands by setting lang=spice. 

• The next line says that node ‘0’ is a global ground signal. We generally do not use global 
ground, but other designers count on it inside sub-circuits. We *do* use 0 for ground in the 
main circuit below. And you must use 0 for the ground reference for your circuit for 
SPICE or Spectre to work correctly. 

• The line starting with include command includes the path for the technology file.  This 
technology, which is also being used in layout drawing, is called `tsmc18d' technology. 
There are two model files, one for the pmos and one for the nmos transistors. $CDK_DIR is 
a system variable you have already set when you run the “setup_all” command, so Spectre 
knows where to find these files. 

• Spectre uses “/” at the end of a line (with no spaces after) to mean “continue this command 
on the next line” SPICE uses a “+” on the next line to mean “this line continues the line 
above it” both types of line continuation work for Spectre. 

• The circuit and sub circuit components in Spectre use “(“ and “)” parenthesis around the 
node list for each component. This is different than SPICE which just lists them.  

• The subcircuit above defines an inverter with two transistors. Note the sizes and other 
parameters. Note the way that signals are “passed” into the subcircuit. 

• After the subcirccuit definitions are the top level circuit components. This shows the 
invocation or instantiation of the subcircuit along with the voltage sources.  

• The rest of the file consists of options and commands to Spectre. The only line you would 
normally change is the “tran” line which sets up the transient analysis.  

• Generally, you would not change the rest of the file, and just copy that code to any new file 
you made by hand.  

• Note: Do not forget that first line of every SPICE (and Spectre) file MUST be a 
comment. 

1.7 Additional information 
To generate the netlist file, Cadence or MMI tools (VLSI design packages) can be used. The circuit 
is designed in these schematics/layout editing tools by using the built-in commands of the tools to 
create a netlist.  The simulations are run and resulting waveforms are observed for circuit 
functionality.  For this assignment, the SPICE files will be written from scratch (using any text 
editor), as one of the objectives of this assignment is to practice writing full-custom SPICE codes to 
represent circuits. NOTE: SPICE code requires all nodes of a circuit to be connected (to a net).  No 
floating nodes are allowed. 
 



2 Setting up your account 
In this lab, you will be using software provided by a company called Cadence.  Spectre will be used 
to simulate the circuit designs. Spectre is Cadence’s proprietary implementation of the popular 
simulator SPICE that was originally developed at the University of California, Berkeley. Other 
versions of SPICE are Pspice, and Hspice. Spectre has many unique features, but we will mostly use 
the features common to all versions of SPICE.  The work will be done in the virtuoso design 
environment which integrates schematic capture, layout and simulation. 
 

3 The assignment 
1. Test the delay of an inverter vs the fanout of the inverter. Use a fan-out of 1, 2, 3, and 4. 

• The Nmos device should be minimum size and the Pmos device should be chosen 
by you to make the rise and fall times as close to each other as possible. Note, 
virtuoso will not let you pick arbitrary sizes. 

• The input to the device under test should be an inverter of equal size to the DUT. 
That inverter should be driven with a source having rising and falling edges of 
10ps. 

• Make a plot of fanout vs delay.  
2. Design a chain of inverters of increasing drive strength to drive a final load capacitor of 

1pf. The first inverter in the chain should be minimum size (as above) and the number 
and sizes of the inverters in the chain are up to you. You want to drive the load as fast 
as possible with as few inverters as possible. The input should be a source with 10ps 
rise and fall times.  
• Record all your design attempts in your report – numbers, sizes, results 
• How do you know you can not do better than the result you have found? 

3. Create NAND and NOR circuits by editing one of your inverter netlists from above by 
editing the code by hand and running Spectre to verify the correctness of the circuit. 
• You will need to add more transistors  to the inverter circuit to convert it into a 

nand  and again for a nor gate.  
• You will have to add a second source for the second input to the logic gates in order 

to generate a “truth table in time”  
 

4. Reports are due by next Friday, September 19, 2008. 
 


